
PROGRAMA GENERAL DE PRUEBAS DE RANGO MÚLTIPLE 
 
C PROGRAMA PRINCIPAL. 
C  
C FWE,PFE Y PCE EN LAS PRUEBAS TRADICIONALES PROTEGIDAS Y  
C NO PROTEGIDAS. 
C 
C ARCHIVO: FWPCPE.FOR 
C REVISION: 23 DE NOVIEMBRE DE 2003 
C 
C PROGRAMADO POR: PAOLA RUELAS VILLEGAS. 
C                 ESTEBAN BURGUETE. 
C                 JOSE FRANCISCO TAMBORERO. 
C  
C PROGRAMA PARA CALCULAR FWE, PCE Y PFE DE LAS PRUEBAS   
C TRADICIONALES PROTEGIDAS Y NO PROTEGIDAS EN EL MODELO  
C COMPLETAMENTE AL AZAR. LOS ERRORES SE ASUMEN  
C INDEPENDIENTES E IDENTICAMENTE DISTRIBUIDOS NORMALES  
C CON MEDIA CERO Y VARIANZA COMUN SIGMA CUADRADA.  
C 
C SE HACE USO DEL METODO DE PLUG IN BOOTSTRAP O BOOTSTRAP 
C PARAMETRICO. ESTE METODO REQUIERE LA SIMULACION DEL MODELO 
C ORIGINAL REDUCIDO POR LA HIPOTESIS NULA, PERO CONSIDERANDO  
C LOS PARAMETROS RESTANTES. VER LA TESIS: 
C 
C 
C        "ESTIMACION DE FWE, PCE Y PFE DE LA PRUEBA DE 
C         EN EL MODELO COMPLETAMENTE AL AZAR."  
C         DEPARTAMENTO DE INGENIERIA INDUSTRIAL.  
C         UNIVERSIDAD DE LAS AMERICAS-PUEBLA. 
C 
C 
C SE HACE USO  
C DE LA SIMULACION. LA SIMULACION CONSIDERA UN PARAMETRO 
C QUE ES: EL NUMERO DE SIMULACIONES(IB) QUE 
C REPRESENTA EL NUMERO DE VECES QUE SE SIMULA UN CONJUNTO DE  
C DATOS CON UN CIERTO NUMERO DE TRATAMIENTOS Y DE REPETICIONES,  
C CON DISTRIBUCION NORMAL Y MEDIAS IGUALES Y VARIANZA  
C COMÚN SIGMA CUADRADA. 
C 
C LA VARIANZA SE TOMA IGUAL A UNO Y LAS MEDIA COMO CERO. ASI, 
C LA SIMULACION SOLO CONTEMPLA GENERAR DATOS iidN(0,1). 
C CABE MENCIONAR QUE EL USUARIO PUEDE FIJAR EL NUMERO DE  
C SIMULACIONES, SE RECOMIENDA USAR COMO MINIMO 1,000,000 DE 
C SIMULACIONES. CON ESTO SE ASEGURA OBTENER RESULTADOS  
C CONFIABLES.  
C 
C 
C COMO LINEAS DE PROGRAMACION, SE TIENEN: 
C 
C             1. EL LENGUAJE DE PROGRAMACION USADO ES EL 
C                FORTRAN 
C             2. LAS VARIABLES Y FUNCIONES SERAN ENTERAS SI 
C                PRINCIPIAN CON LAS LETRAS I,J,K,L,M,N. Y  
C                SERAN REALES SI PRINCIPIAN CON CUALQUIER  
C                OTRA LETRA. 
C             3. DEBIDO A PROBLEMAS CON EL RECONOCIMIENTO DE 
C                ALGUNOS CARACTERES, SE HA DECIDIDO NO USAR  
C                ACENTOS. POR LO MISMO, SE DECIDIO USAR  

C                MAYUSCULAS EN LOS COMENTARIOS. EN EL PROGRAMA 
C                PUEDEN APARECER LINEAS EN MINUSCULAS. 
C             4. EL PROGRAMA TRABAJA INTERACTIVAMENTE.  
C 
C SE DEFINEN ALGUNOS PARAMETROS 
C 
      CHARACTER*12 SALIDA 
      COMMON C0,C1,C2,D1,D2,D3 
      COMMON SALIDA 
      COMMON IB 
      CHARACTER*80 TITULO 
C 
C A CONTINUACION SE ASIGNAN LAS CONSTANTES                 
C QUE SE UTILIZAN PARA GENERAR 
C LOS NUMEROS ALEATORIOS NORMALES CON MEDIA CERO Y  
C VARIANZA UNO.   
C  
C PARA MAYOR INFORMACION VER LA FORMULA  
C 26.1 DE: 
C 
C        ABAMOWITZ,M. & STEGUN, I.A.(1964)  
C        HANDBOOK OF MATHEMATICAL FUNCTIONS. 
C        NATIONAL BUREAU OF STANDARDS.  
C        WASHINGTON,DC. 
C 
      C0=2.515517       
      C1=0.802853 
      C2=0.010328 
      D1=1.432788 
      D2=0.189269 
      D3=0.001308 
C  
C AQUI SE LEE EL NOMBRE DEL ARCHIVO DE  
C SALIDA. 
C EL ARCHIVO DE SALIDA LO CONSTRUYE EL PROGRAMA. SI YA  
C EXISTE UN ARCHIVO CON ESE NOMBRE, POR PROTECCION, 
C EL PROGRAMA SE DETIENE CON UN MENSAJE DE ERROR.   
C 
    1 FORMAT(A12) 
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NOMBRE DEL ARCHIVO DE SALIDA(OUTPUT)' 
      READ(*,1)SALIDA 
      OPEN(5,FILE=SALIDA,STATUS='NEW')       
C 
C 
C AQUI SE LEE EL TITULO. ESTE DEBE SER UNA LINEA DE 
C NO MAS DE 80 CARACTERES, ESTE TITULO SE PONDRA EN EL 
C ARCHIVO DE SALIDA. 
C 
    2 FORMAT(A80) 
      PRINT*,'  ' 
      PRINT*,' INTRODUCE EL TITULO(UNA LINEA < 80 CARACTERES)' 
      READ(*,2)TITULO 
      WRITE(5,2)TITULO    
C       
C SE PONE EL MENSAJE INICIAL EN EL ARCHIVO DE SALIDA.       
C  
      WRITE(5,179) 
  179 FORMAT(/ 
     110X,'PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE DE '/ 

     210X,'LAS PRM (SI Y NO) PROTEGIDAS EN EL '/ 
     310X,'DISEÑO COMPLETAMENTE AL AZAR,'// 
     510X,'PROGRAMADO POR: PAOLA RUELAS VILLEGAS.'/ 
     610X,'                     ESTEBAN BURGUETE.'/ 
     710X,'                       JOSE TAMBORERO.'/) 
C 
C SE MANDA EL NOMBRE DEL ARCHIVO DE SALIDA AL 
C ARCHIVO DE SALIDA.         
C               
      WRITE(5,180)SALIDA 
  180 FORMAT(3(/),10X,'ARCHIVO DE  SALIDA: ',A12)     
C 
C SE PIDE QUE SE INTRODUZCA LA SEMILLA. 
C 
   34 PRINT*,' '      
      PRINT*,'   INTRODUCE LA SEMILLA:'  
      PRINT*,'   DEBE SER UN ENTERO MAYOR QUE  0' 
      PRINT*,'   Y MENOR QUE 2147483647' 
      READ*,ISEMILLA 
      IF(ISEMILLA.LE.0.OR.ISEMILLA.GE.2147483647)GO TO 34 
C 
C LA SEMILLA SE GRABA EN EL ARCHIVO DE SALIDA. 
C       
      WRITE(5,181)ISEMILLA 
  181 FORMAT(//10X,'SEMILLA PARA SIMULACION: ',I10) 
C 
C  
C POR SEGURIDAD, SE DESHECHAN LOS PRIMEROS 1000 NUMEROS  
C GENERADOS POR LA SEMILLA. ESTO SE HACE PARA ALEATORIZAR  
C MEJOR LA SEMILLA, YA QUE POR CADA NUMERO QUE SE GENERA, SE 
C OBTIENE UNA NUEVA SEMILLA. 
C 
      DO 5 II=1,1000 
      YY=UNIFO(ISEMILLA) 
    5 CONTINUE 
C 
C 
C SE FIJA EL NUMERO DE SIMULACIONES(IB). 
C                                            
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NUMERO DE SIMULACIONES' 
      READ*,IB      
C 
C ESTA INFORMACION SE MANDA AL ARCHIVO DE SALIDA. 
C 
      WRITE(5,188)IB 
  188 FORMAT(//10X,'NUMERO TOTAL DE SIMULACIONES: ',I10) 
C 
C       
C EN ESTA PARTE SE PREGUNTA SI SE DESEA CORRER OTRA 
C SIMULACION. SI SE CONTESTA CON "1" ES "SI", 
C PARA EL "NO" PUEDE PONERSE CUALQUIER OTRO NUMERO.  
C SI SE METE UN CARACTER ALFABETICO,  
C EL PROGRAMA TERMINA CON UN MENSAJE DE ERROR. 
C       
  300 CONTINUE 
      PRINT*,'  '       
      PRINT*,'   DESEAS CORRER UNA SIMULACION?(SI=1)' 
      READ(*,*)IDESEA 
      IF(IDESEA.NE.1)GO TO 200 

C 
C 
C AQUI SE HACE LA LLAMADA A LA SUBRUTINA QUE HACE  
C LAS ESTIMACIONES DEL FWE, PCE Y PFE DE LAS PRUEBAS DE 
C RANGO MULTIPLE EN EL DISEÑO COMPLETAMENTE AL AZAR,  
C PROTEGIDAS Y NO PROTEGIDA. 
C 
C 
      CALL FWPCPE(ISEMILLA) 
C 
C  
C SE REGRESA A PREGUNTAR SI SE DESEA ESTIMAR  
C EL FWE, PCE Y PFE EN OTRA COMBINACION DE  
C TRATAMIENTOS Y REPETICIONES. 
C  
      GO TO 300 
C 
C SI NO HAY OTRA SIMULACION POR REALIZAR, LA  
C INSTRUCCIÓN DEL PRINCIPIO SALTARA AL 200 CONTINUE 
C QUE SE ENCUENTRA AQUI ABAJO, CON LO CUAL SE IMPRIME 
C UN MENSAJE EN EL ARCHIVO DE SALIDA Y SE TERMINA LA  
C EJECUCION. 
C       
  200 CONTINUE 
C 
C MENSAJE DE FIN DEL PROGRAMA AL ARCHIVO DE SALIDA. 
C   
      WRITE(5,184) 
C 
C SE CIERRA EL ARCHIVO DE SALIDA. 
C  
      CLOSE(5)   
C 
C UN MENSAJE FINAL DE FIN DEL PROGRAMA SE PONE EN  
C PANTALLA. 
C 
 
      WRITE(*,184) 
  184 FORMAT(///20X,'           FIN DEL PROGRAMA')     
C 
C 
C SE TERMINA LA EJECUCION. 
C       
      STOP 
      END 
C 
C       
      SUBROUTINE FWPCPE(ISEMILLA) 
C 
C  
C SUBRUTINA CALCULAR EL FWE, PCE Y PFE DE LAS  
C PRUEBAS TRADICIONALES DE RANGO MULTIPLE  
C PROTEGIDAS Y NO PROTEGIDAS EN UN 
C DISEÑO COMPLETAMENTE AL AZAR. 
C   
      dimension rmydata(20,20) 
      dimension RMED(20),VALCRIT(20) 
      DIMENSION COMP(20) 
      REAL CME 
      CHARACTER*12 SALIDA 

      COMMON C0,C1,C2,D1,D2,D3 
      COMMON SALIDA 
      COMMON IB 
C 
C 
C SE LEE lA INFORMACION RELACIONADA CON EL NUMERO 
C DE TRATAMIENTOS Y REPETICIONES. 
C 
      PRINT*,'  ' 
      PRINT*,'   INTRODUCE EL NUMERO DE TRATAMIENTOS' 
      PRINT*,'   Y DE REPETICIONES'                             
      READ*,NTRAT,NREP   
C 
C 
C SE CALCULA EL NUMERO DE COMPARACIONES (N ESCOGE 2, ES 
C DECIR COMBINACIONES DE N TOMANDO 2 A LA VEZ). 
C 
      RNTRAT=NTRAT 
      RNCOMP=RNTRAT*(RNTRAT-1.)/2. 
      NCOMP=RNCOMP 
C 
C 
C SE CALCULAN LOS GRADOS LIBRES DEL ERROR. 
C 
      RNREP=NREP 
      GLE=RNTRAT*(RNREP-1.) 
C 
C 
C SE PIDE QUE SE INTRODUZCAN LOS VALORES   
C CRITICOS. ESTOS VALORES EN EL PRESENTE TRABAJO NO 
C SON SIEMPRE LOS QUE SE SACAN DE LA TABLA. PARA HACER 
C POSIBLE ESTE PROGRAMA, FUE NECESARIO ESCRIBIR  
C LAS PRUEBAS DE RANGO MULTIPLE EN LA FORMA 
C 
C PARA PROBAR Ho:M(I)=M(J),  
C 
C LA REGLA DE DECISION ES: RECHAZAR Ho. SI: 
C                                   _ 
C         ABS(Y(I)-Y(J)) >= K(I,J)*S(Y) 
C  
C DONDE LA K(I,J) REPRESENTA EL VALOR CRITICO PARA 
C LA COMPARACION DE LA MEDIA I CONTRA LA J, ORDENADAS. 
C           
C ALGUNOS CALCULOS PUEDEN SER NECESARIOS ANTES DE 
C UTILIZAR ESTE PROGRAMA. SI SE DESEARA POR EJEMPLO 
C HACER LAS ESTIMACIONES EN LA PRUEBA DMS, ES NECESARIO  
C MULTIPLICAR LAS T'S QUE SE OBTIENEN DE LAS TABLAS  
C POR SQRT(2). PERO PARA LA PRUEBA DE TUKEY, NO ES  
C NECESARIO. 
C  
C ESTE PROGRAMA SE DISEÑO SOLAMENTE PARA LA  
C COMPARACION CON ALFA=0.05, ALGO MAS DE PROGRAMACION 
C DEBE HACERSE PARA HACERLO MAS GENERAL. SE USO EL 
C ALFA MENCIONADO PORQUE ES EL MAS COMUNMENTE 
C ENCONTRADO. 
C 
C AHORA SE PIDE QUE SE METAN LOS VALORES CRITICOS PARA 
C LAS COMPARACIONES.  
C 
      DO 500 J=2,NTRAT 

      PRINT*,' VALOR CRITICO 1 VS ',J 
      READ*,VALCRIT(J) 
  500 CONTINUE 
C 
C  
C EL NUMERO DE TRATAMIENTOS, REPETICIONES Y  
C COMPARACIONES SE MANDAN AL ARCHIVO DE SALIDA. 
C 
      WRITE(5,139)NTRAT,NREP,NCOMP 
  139 FORMAT(//10X,'NUMERO DE TRATAMIENTOS:',I4//     
     1         10X,'NUMERO DE REPETICIONES:',I4// 
     1         10X,'NUMERO DE COMPARACIONES:',I4) 
C 
C 
C EL VALOR DE LAS R'S DE DUNCAN SE MANDAN A  
C IMPRIMIR AL ARCHIVO DE SALIDA. 
C                            
      WRITE(5,141)GLE 
  141 FORMAT(//10X,' VALORES CRITICOS CON ',F7.3, 
     1' GRADOS LIBRES.') 
      DO 488 I=2,NTRAT 
      WRITE(5,140)I,VALCRIT(I) 
  140 FORMAT(//10X,' CON ',I3,' MEDIAS INVOLUCRADAS: ', 
     1F7.3) 
  488 CONTINUE    
C 
C 
C SE PIDE LA F DE TABLAS, CON NTRAT-1 Y  
C NTRAT*(NREP-1) GRADOS DE LIBERTAD. 
C 
      PRINT*,' INTRODUCE LA F CON ',NTRAT-1,' Y ', 
     1GLE,' GRADOS LIBRES.' 
      READ*,FTABLAS 
C 
C 
C LA INFORMACION DE LA F DE TABLAS SE MANDA AL  
C ARCHIVO DE SALIDA. 
C 
      WRITE(5,421)NTRAT-1,GLE,FTABLAS 
  421 FORMAT(//10X,'F DE TABLAS CON ',I3,' Y ',F6.2, 
     1'GRADOS LIBRES= ',F6.2) 
C 
C 
C SE INICIALIZAN LOS CONTADORES DE RECHAZOS. 
C 
C  IPFENP=CONTADOR PER FAMILY NO PROTEGIDA 
C  IFWENP=CONTADOR FAMILYWISE NO PROTEGIDA 
C  IPFEP=CONTADOR PER FAMILY PROTEGIDA 
C  IPFWP=CONTADOR FAMILYWISE PROTEGIDA 
C 
      IPFENP=0 
      IFWENP=0 
      IPFEP=0 
      IFWEP=0 
C 
C 
C EN ESTE LUGAR SE DA INICIO A LA SIMULACION. 
C IB=NUMERO DE SIMULACIONES. 
C                                   
      do 322 nxx=1,IB 

C 
C 
C SE GENERAN DATOS DE UNA NORMAL CON MEDIA CERO Y  
C VARIANZA UNO. LOS DATOS SE GUARDAN EN LA MATRIZ 
C RMYDATA(20,20). ESTOS DATOS SON COMO DATOS  
C OBSERVADOS. 
C 
      DO 883 I=1,NTRAT 
      DO 883 J=1,NREP 
      Y=UNIFO(ISEMILLA) 
      RMYDATA(I,J)=ZQNT(Y) 
  883 CONTINUE 
C 
C 
C SE LLAMA A LA SUBRUTINA PARA CALCULAR EL CUADRADO  
C MEDIO DEL ERROR, LAS MEDIAS DE LOS TRATAMIENTOS Y 
C LA F CALCULADA. 
C      
      FC=0.0 
      CME=0.0 
      call resca2(ntrat,nrep,rmydata,RMED,cme,FC) 
C 
C 
C SE CALCULAN LOS ESTADISTICOS DE DUNCAN PARA LA 
C SIMULACION ACTUAL. 
C      
      DO 33 I=2,NTRAT 
   33 COMP(I)=VALCRIT(I)*SQRT(CME/RNREP) 
C 
C SE ORDENAN LAS MEDIAS DE LOS TRATAMIENTOS. 
C                                            
      CALL SOLORD(RMED,NTRAT) 
C 
C SE CALCULA EL NUMERO DE RECHAZOS PARA LA SIMULACION. 
C       
      ICONT=0 
      DO 590 I=1,NTRAT-1 
      DO 590 J=I+1,NTRAT 
      KK=J-I+1 
      IF(RMED(J)-RMED(I).GT.COMP(KK))ICONT=ICONT+1 
  590 CONTINUE 
C 
C 
C AQUI SE GUARDAN ESTOS RESULTADOS 
C PARA LOS CONTADORES DEL FWE Y PFE PARA LA PRUEBA 
C NO PROTEGIDA 
C . 
C 
      IPFENP=IPFENP+ICONT 
      IF(ICONT.GE.1)IFWENP=IFWENP+1 
C 
C 
C AQUI SE GUARDAN LOS RESULTADOS PARA LOS CONTADORES 
C DEL FWE Y PFE DE LA PRUEBA PROTEGIDA. 
C 
      IF(FC.LT.FTABLAS)GO TO 322 
      IPFEP=IPFEP+ICONT 
      IF(ICONT.GE.1)IFWEP=IFWEP+1 
C 
C 

C AQUI SE TERMINA EL LOOP DE LAS SIMULACIONES. 
C                                                          
  322 CONTINUE 
C 
C 
C EN ESTA PARTE SE CUANTIFICAN LOS ERRORES, SE  
C ESTABLECEN LOS INTERVALOS DE CONFIANZA Y SE  
C IMPRIMEN EN EL ARCHIVO DE SALIDA. 
C 
C 
C PRUEBA NO PROTEGIDA. 
C 
C 
C PRIMERAMENTE SE RESUELVE EL PROBLEMA PARA EL 
C FWE. EL FWE PUEDE CONCEBIRSE COMO LA PROPORCION 
C DE VECES QUE HUBO POR LO MENOS UN RECHAZO. EN 
C ESTE PROCESO, UN RECHAZO ES UNA COMPARACION  
C ERRONEA. 
C 
C     RECORDAR:   
C               RIB= NUMERO DE SIMULACIONES. 
C            RIREPB= NUMERO DE REPETICIONES POR SIMULACION. 
C            RNCOMP= NUMERO DE COMPARACIONES. 
C 
      RIB=IB 
      RNCOMP=NCOMP 
C 
C INFERENCIA PARA FWE. 
C 
      RFWENP=IFWENP 
      RFWENP=RFWENP/RIB 
C 
C 
C AHORA SE RESUELVE EL PROBLEMA PARA EL PFE. EL FWE 
C PUEDE VERSE COMO EL NUMERO PROMEDIO DE RECHAZOS.  
C YA QUE SE TRATA DE UNA MEDIA, PUEDE ARGUMENTARSE 
C QUE EL TEOREMA DEL LIMITE CENTRAL ES APLICABLE. 
C 
      RPFENP=IPFENP 
      RPFENP=RPFENP/RIB 
C 
C 
C EN ESTA PARTE SE REALIZA LA INFERENCIA PARA  
C EL PCE. POR DEFINICION, EL PCE ES IGUAL AL 
C PFE DIVIDIDO POR EL NUMERO DE COMPARACIONES. 
C POR LO QUE USANDO ESTA PROPIEDAD, SE OBTIENE 
C EL INTERVALO DE CONFIANZA. 
C                            
      RPCENP=RPFENP/RNCOMP 
C      ZT=0 
C 
C 
C AQUI SE IMPRIMEN FWE,PFE Y PCE, DE LA PRUEBA 
C NO PROTEGIDA EN EL ARCHIVO DE SALIDA. 
C 
 
      WRITE(5,864)RFWENP,RPFENP,RPCENP 
  864 FORMAT(// 10X,' PRUEBA NO PROTEGIDA', 
     1       //10X,' FWE: ',F12.6, 
     1       //10X,' PFE: ',F12.6, 

     1       //10X,' PCE: ',F12.6) 
C 
C PRUEBA PROTEGIDA. 
C 
C 
C PRIMERAMENTE SE RESUELVE EL PROBLEMA PARA EL 
C FWE. EL FWE PUEDE CONCEBIRSE COMO LA PROPORCION 
C DE VECES QUE HUBO POR LO MENOS UN RECHAZO. EN 
C ESTE PROCESO, UN RECHAZO ES UNA COMPARACION  
C ERRONEA. 
C 
C     RECORDAR:   
C               RIB= NUMERO DE SIMULACIONES. 
C            RIREPB= NUMERO DE REPETICIONES POR SIMULACION. 
C            RNCOMP= NUMERO DE COMPARACIONES. 
C 
      RIB=IB 
      RNCOMP=NCOMP 
C 
C INFERENCIA PARA FWE. 
C 
      RFWEP=IFWEP 
      RFWEP=RFWEP/RIB 
C 
C 
C AHORA SE RESUELVE EL PROBLEMA PARA EL PFE. EL FWE 
C PUEDE VERSE COMO EL NUMERO PROMEDIO DE RECHAZOS.  
C YA QUE SE TRATA DE UNA MEDIA, PUEDE ARGUMENTARSE 
C QUE EL TEOREMA DEL LIMITE CENTRAL ES APLICABLE. 
C 
      RPFEP=IPFEP 
      RPFEP=RPFEP/RIB 
C 
C 
C EN ESTA PARTE SE REALIZA LA INFERENCIA PARA  
C EL PCE. POR DEFINICION, EL PCE ES IGUAL AL 
C PFE DIVIDIDO POR EL NUMERO DE COMPARACIONES. 
C                            
      RPCEP=RPFEP/RNCOMP 
C      ZT=0 
C 
C 
C AQUI SE IMPRIMEN FWE,PFE Y PCE, DE LA PRUEBA  
C PROTEGIDA EN EL ARCHIVO DE SALIDA. 
C 
 
      WRITE(5,863)RFWEP,RPFEP,RPCEP 
  863 FORMAT(// 10X,' PRUEBA PROTEGIDA', 
     1       //10X,' FWE: ',F12.6, 
     1       //10X,' PFE: ',F12.6, 
     1       //10X,' PCE: ',F12.6,//    
     210X,'______________________________')      
C 
C SE REGRESA AL PROGRAMA PRINCIPAL. 
C  
      RETURN 
      END 
c 
c 
      subroutine resca2(ntrat,nrep,a,t,c,F) 

c 
C 
c Esta subrutina sirve para calcular LAS MEDIAS de 
c tratamientos y el cuadrado medio del error en un diseño 
c completamente al azar. 
C 
c Argumentos: NTRAT=NUMERO DE TRATAMIENTOS. 
c             NREP=NUMERO DE REPETICIONES. 
c             A=MATRIZ DE OBSERVACIONES. 
c 
c     Salida: 
c             t=VECTOR DE LAS MEDIAS DE LOS TRATAMIENTOS. 
c             c=cuadrado medio del error 
C             F=F CALCULADA 
c   
      dimension a(20,20) 
      dimension t(20) 
C 
C INICIALIZACION DE LAS SUMAS NECESARIAS PARA 
C CALCULAR LAS MEDIAS Y EL CUADRADO MEDIO DEL ERROR. 
C       
      sum=0 
      sum2=0 
      sum2t=0.0 
C 
C LOS ENTEROS NREP Y NTRAT SE CONVIERTEN EN REALES 
C PARA CALCULOS POSTERIORES. 
C 
      RNREP=NREP             
      RNTRAT=NTRAT 
C 
C SUMA SOBRE TRATAMIENTOS E INICIALIZACION DEL 
C VECTOR DE MEDIAS T(20) 
C       
      do 2 i=1,ntrat 
      t(i)=0 
C 
C SUMA SOBRE LAS REPETICIONES 
C 
 
      do 3 j=1,nrep 
      sum=sum+a(i,j) 
      t(i)=t(i)+a(i,j) 
      sum2=sum2+a(i,j)*a(i,j) 
    3 continue 
      sum2t=sum2t+t(i)*t(i) 
      T(I)=T(I)/RNREP 
    2 continue 
C 
C SE CALCULA EL CUADRADO MEDIO DEL ERROR 
C 
      CMTRAT=(SUM2T/RNREP-SUM*SUM/(RNREP*RNTRAT))/(RNTRAT-1.) 
      c=(sum2-sum2t/rnrep)/(rntrat*(rnrep-1.0)) 
      F=CMTRAT/C 
      return 
      end 
C 
C 
      FUNCTION UNIFO(IX) 
C 

C 
C GENERADOR DE NUMEROS ALEATORIOS IMPLEMENTANDO LOS RECURSOS: 
C IX=16807*IX MOD (2**(31)-1) 
C USANDO SOLO 32 BITS, ICLUYENDO SIGNO. 
C 
C ALGUNOS COMPILADORES REQUIEREN LA DECLARACION: 
C INTEGER*4 IX, K1 
C  
C INPUT 
C IX= ENTERO MAYOR QUE 0 Y MENOR QUE 2147483647     
C 
C OUTPUT 
C IX= NUEVO VALOR PSEUDO-ALEATORIO, 
C UNIF= UNA FRACCION UNIFORME ENTRE 0 Y 1    
C 
C LA FUNCION GENERADORA DE NUMEROS ALEATORIOS 
C      
      K1=IX/127773 
      IX=16807*(IX-K1*127773)-K1*2836 
      IF(IX.LE.0.OR.IX.GE.2147483647)IX=IX+2147483647 
      UNIFO=IX*4.65661287e-10  
      RETURN 
      END 
C 
C           
      FUNCTION ZQNT(UNIF)  
C 
C 
C ESTA FUNCION EVALUA LOS CUANTILES 
C DE LA DISTRIBUCION NORMAL ESTANDAR 
C LAS CONSTANTES, C0,C1,C2 Y D1,D2,D3, DEBEN  
C SER DEFINIDAS EN EL PROGRAMA PRINCIPAL A SABER:  
C 
C      C0=2.515517 
C      C1=0.802853 
C      C2=0.010328 
C      D1=1.432788 
C      D2=0.189269 
C      D3=0.001308 
C 
C PARA MAYOR INFORMACION VER LA  
C FORMULA 26.1 DE: 
C 
C        ABAMOWITZ & STEGUN.  
C        A HANDBOOK OF MATHEMATICAL FUNCTIONS. 
C        NATIONAL BUREAU OF STANDARDS.  
C        WASHINGTON,DC. 
C     
      COMMON C0,C1,C2,D1,D2,D3 
      PP=UNIF 
      MULT=1 
      IF(PP.LT.0.5)MULT=-1 
      IF(PP.GT.0.5)PP=1-PP 
      if(pp.lt.1e-30)pp=1e-30 
      T=SQRT(LOG(1./(PP*PP))) 
      ZQNT=(T-((((C2*T+C1)*T)+C0)/(1+((((D3*T+D2)*T)+ 
     1D1)*T))))*MULT 
      IF(PP.EQ.0.5)ZQNT=0.0 
      RETURN 
      END 

C    
C 
      SUBROUTINE SOLORD(RMORD,NTRAT)     
C 
C     
C ESTA SUBRUTINA SIRVE PARA ORDENAR LAS MEDIAS  
C DE LOS TRATAMIENTOS EN FORMA ASCENDENTE.  
C LAS MEDIAS VIENEN EN EL VECTOR RMORD(20). LOS DATOS  
C QUEDARAN ORDENADOS EN EL MISMO VECTOR RMORD(20) 
C QUE SE RECIBIO COMO ARGUMENTO. 
C 
      DIMENSION RMORD(20)       
      DO 133 I=1,NTRAT-1 
      DO 133 J=I+1,NTRAT 
      IF(RMORD(I).LE.RMORD(J)) GO TO 133 
      T=RMORD(I) 
      RMORD(I)=RMORD(J) 
      RMORD(J)=T 
  133 CONTINUE 
      RETURN 
      END 

CORRIDA DEL PROGRAMA GENERAL DE PRUEBAS DE RANGO 
MÚLTIPLE 
 

 DMS 
 

PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE  
DE LAS PRM (SI Y NO) PROTEGIDAS EN EL  

     DISEÑO COMPLETAMENTE AL AZAR 
 
 
          PROGRAMADO POR:    PAOLA RUELAS       
                              ESTEBAN BURGUETE. 
                              JOSE TAMBORERO. 
 
 
          ARCHIVO DE  SALIDA: DMSI.LIS      
 
 
          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 
 
     TIPO DE PRUEBA: DMS 

 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 
 
VALOR CRITICO 
  VALOR CRITICO 1 VS 2: 3.014 

  VALOR CRITICO 1 VS 3: 3.014 

  VALOR CRITICO 1 VS 4: 3.014 

  VALOR CRITICO 1 VS 5: 3.014 

   
 
           PRUEBA DMS NO PROTEGIDA 
 
           FWE:      .257485 
 
           PFE:      .500692 
 
           PCE:      .050069 
 
 
           PRUEBA DMS PROTEGIDA 
 
           FWE:      .049972 

           PFE:      .179178 
 
           PCE:      .017918 
 
          _____________________________ 
 
 
                               FIN DEL PROGRAMA 
 
 

 TUKEY 
 
          PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE DE  
          LA PRUEBA DE TUKEY (SI Y NO) PROTEGIDA EN EL  
          DISEÑO COMPLETAMENTE AL AZAR, 
 
          PROGRAMADO POR:     PAOLA RUELAS      
                              ESTEBAN BURGUETE. 
                                JOSE TAMBORERO. 
 
 
 
 
          ARCHIVO DE  SALIDA: TUKEYI.LIS    
 
 
          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 

 

          TIPO DE PRUEBA: TUKEY 

 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 
 
VALOR CRITICO 
  VALOR CRITICO 1 VS 2: 4.37 

  VALOR CRITICO 1 VS 3: 4.37 

  VALOR CRITICO 1 VS 4: 4.37 

  VALOR CRITICO 1 VS 5: 4.37 

 
 
 
 
           PRUEBA DE TUKEY NO PROTEGIDA 

           FWE:      .049802 
 
           PFE:      .075050 
 
           PCE:      .007505 
 
 
           PRUEBA DE TUKEY PROTEGIDA 
 
           FWE:      .041636 
 
           PFE:      .066884 
 
           PCE:      .006688 
 
          ______________________________ 
 
 
 
                               FIN DEL PROGRAMA 
 

 SNK 
           

PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE DE  
          LA PRUEBA DE TUKEY (SI Y NO) PROTEGIDA EN EL  
          DISEÑO COMPLETAMENTE AL AZAR, 
 
          PROGRAMADO POR:     PAOLA RUELAS      
                              ESTEBAN BURGUETE. 
                                JOSE TAMBORERO. 
 
 
 
 
          ARCHIVO DE  SALIDA: SNKI.LIS    
 
 
          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 

     TIPO DE PRUEBA: SNK 

 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 

VALORES CRITICOS  
 
  VALOR CRITICO 1 VS 2: 3.010 

  VALOR CRITICO 1 VS 3: 3.670 

  VALOR CRITICO 1 VS 4: 4.080 

  VALOR CRITICO 1 VS 5: 4.370 

 
 
           PRUEBA DE SNK NO PROTEGIDA 
 
           FWE:      .054938 
 
           PFE:      .112608 
 
           PCE:      .011261 
 
 
           PRUEBA DE SNK PROTEGIDA 
 
           FWE:      .044296 
 
           PFE:      .101404 
 
           PCE:      .010140 
 
          ______________________________ 
 
 
 
                               FIN DEL PROGRAMA 
 

 
 DUNCAN 
 
   PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE DE  

          LA PRUEBA DE TUKEY (SI Y NO) PROTEGIDA EN EL  
          DISEÑO COMPLETAMENTE AL AZAR, 
 
          PROGRAMADO POR:     PAOLA RUELAS      
                              ESTEBAN BURGUETE. 
                                JOSE TAMBORERO. 
 
 
 
 
          ARCHIVO DE  SALIDA: DUNCANI.LIS    
 
 
          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 

     TIPO DE PRUEBA: DUNCAN 

 
          NUMERO DE TRATAMIENTOS:   5 

          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 
 
  VALORES CRITICOS  
 
  CON 2 MEDIAS INVOLUCRADAS: 3.010 

  CON 3 MEDIAS INVOLUCRADAS: 3.160 

  CON 4 MEDIAS INVOLUCRADAS: 3.250 

  CON 5 MEDIAS INVOLUCRADAS: 3.310 

 
 
           PRUEBA DE DUNCAN NO PROTEGIDA 
 
           FWE:      .186100 
 
           PFE:      .361420 
 
           PCE:      .036142 
 
 
           PRUEBA DE DUNCAN PROTEGIDA 
 
           FWE:      .049216 
 
           PFE:      .162270 
 
           PCE:      .016227 
 
          ______________________________ 
 
 
 
                               FIN DEL PROGRAMA 
 
 
 

 BONFERRONI 
 

PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE  
DE LAS PRM (SI Y NO) PROTEGIDAS EN EL  

     DISEÑO COMPLETAMENTE AL AZAR 
 
 
          PROGRAMADO POR:    PAOLA RUELAS       
                              ESTEBAN BURGUETE. 
                              JOSE TAMBORERO. 
 
 
          ARCHIVO DE  SALIDA: BONFI.LIS      

          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 
 
     TIPO DE PRUEBA: BONFERRON 

 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 
 
VALORES CRITICOS 
 
  VALOR CRITICO 1 VS 2: 4.650 

  VALOR CRITICO 1 VS 3: 4.650 

  VALOR CRITICO 1 VS 4: 4.650 

  VALOR CRITICO 1 VS 5: 4.650 

   
 
           PRUEBA BONFERRON NO PROTEGIDA 
 
           FWE:      .034256 
 
           PFE:      .049656 
 
           PCE:      .004966 
 
 
           PRUEBA BONFERRON PROTEGIDA 
 
           FWE:      .032858 
 
           PFE:      .048258 
 
           PCE:      .004826 
 
          _____________________________ 
 
 
                               FIN DEL PROGRAMA 

 SCHEFFÉ 
 

PROGRAMA PARA CALCULAR EL FWE, PCE Y PFE  
DE LAS PRM (SI Y NO) PROTEGIDAS EN EL  

     DISEÑO COMPLETAMENTE AL AZAR 
 
 
          PROGRAMADO POR:    PAOLA RUELAS       
                              ESTEBAN BURGUETE. 
                              JOSE TAMBORERO. 
 
 
          ARCHIVO DE  SALIDA: SCHEI.LIS      
 
 
          SEMILLA PARA SIMULACION:     98765 
 
 
          NUMERO TOTAL DE SIMULACIONES: 1000000 
 
     TIPO DE PRUEBA: SCHEFFE 

 
          NUMERO DE TRATAMIENTOS:   5 
 
          NUMERO DE REPETICIONES:   4 
 
          NUMERO DE COMPARACIONES:  10 
 
 
           GRADOS LIBRES DEL ERROR: 15.00 
 
 
          F DE TABLAS CON   4 Y  15.00GRADOS LIBRES=   3.06 
 
VALORES CRITICOS 
 
  VALOR CRITICO 1 VS 2: 4.950 

  VALOR CRITICO 1 VS 3: 4.950 

  VALOR CRITICO 1 VS 4: 4.950 

  VALOR CRITICO 1 VS 5: 4.950 

   
 
           PRUEBA SCHEFFE NO PROTEGIDA 
 
           FWE:      .023250 
 
           PFE:      .032888 
 
           PCE:      .003289 
 
 
           PRUEBA SCHEFFE PROTEGIDA 
 
           FWE:      .023250 
 
           PFE:      .032888 

           PCE:      .003289 


